

NAG Fortran Library Routine Document

F08NEF (SGEHRD/DGEHRD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08NEF (SGEHRD/DGEHRD) reduces a real general matrix to Hessenberg form.

2 Specification

```
SUBROUTINE F08NEF(N, ILO, IHI, A, LDA, TAU, WORK, LWORK, INFO)
ENTRY      sgehrd (N, ILO, IHI, A, LDA, TAU, WORK, LWORK, INFO)
INTEGER    N, ILO, IHI, LDA, LWORK, INFO
real      A(LDA,*), TAU(*), WORK(*)
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine reduces a real general matrix A to upper Hessenberg form H by an orthogonal similarity transformation: $A = QHQ^T$.

The matrix Q is not formed explicitly, but is represented as a product of elementary reflectors (see the F08 Chapter Introduction for details). Routines are provided to work with Q in this representation (see Section 8).

The routine can take advantage of a previous call to F08NHF (SGEBAL/DGEBAL), which may produce a matrix with the structure:

$$\begin{pmatrix} A_{11} & A_{12} & A_{13} \\ & A_{22} & A_{23} \\ & & A_{33} \end{pmatrix}$$

where A_{11} and A_{33} are upper triangular. If so, only the central diagonal block A_{22} , in rows and columns i_{lo} to i_{hi} , needs to be reduced to Hessenberg form (the blocks A_{12} and A_{23} will also be affected by the reduction). Therefore the values of i_{lo} and i_{hi} determined by F08NHF (SGEBAL/DGEBAL) can be supplied to the routine directly. If F08NHF (SGEBAL/DGEBAL) has not previously been called however, then i_{lo} must be set to 1 and i_{hi} to n .

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: N – INTEGER *Input*

On entry: n , the order of the matrix A .

Constraint: $N \geq 0$.

2: ILO – INTEGER *Input*

3: IHI – INTEGER *Input*

On entry: if A has been output by F08NHF (SGEBAL/DGEBAL), then ILO and IHI **must** contain the values returned by that routine. Otherwise, ILO must be set to 1 and IHI to N .

Constraints:

$$1 \leq \text{ILO} \leq \text{IHI} \leq N \text{ if } N > 0, \\ \text{ILO} = 1 \text{ and } \text{IHI} = 0 \text{ if } N = 0.$$

- 4: A(LDA,*) – *real* array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the n by n general matrix A.
On exit: A is overwritten by the upper Hessenberg matrix H and details of the orthogonal matrix Q .
- 5: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08NEF (SGEHRD/DGEHRD) is called.
Constraint: $\text{LDA} \geq \max(1, N)$.
- 6: TAU(*) – *real* array *Output*
Note: the dimension of the array TAU must be at least $\max(1, N - 1)$.
On exit: further details of the orthogonal matrix Q .
- 7: WORK(*) – *real* array *Workspace*
Note: the dimension of the array WORK must be at least $\max(1, \text{LWORK})$.
On exit: if $\text{INFO} = 0$, $\text{WORK}(1)$ contains the minimum value of LWORK required for optimum performance.
- 8: LWORK – INTEGER *Input*
On entry: the dimension of the array WORK as declared in the (sub)program from which F08NEF (SGEHRD/DGEHRD) is called, unless $\text{LWORK} = -1$, in which case a workspace query is assumed and the routine only calculates the optimal dimension of WORK (using the formula given below).
Suggested value: for optimum performance LWORK should be at least $N \times nb$, where nb is the **blocksize**.
Constraint: $\text{LWORK} \geq \max(1, N)$ or $\text{LWORK} = -1$.
- 9: INFO – INTEGER *Output*
On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

$\text{INFO} < 0$

If $\text{INFO} = -i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed Hessenberg matrix H is exactly similar to a nearby matrix $A + E$, where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$ is a modestly increasing function of n , and ϵ is the *machine precision*.

The elements of H themselves may be sensitive to small perturbations in A or to rounding errors in the computation, but this does not affect the stability of the eigenvalues, eigenvectors or Schur factorization.

8 Further Comments

The total number of floating-point operations is approximately $\frac{2}{3}q^2(2q + 3n)$, where $q = i_{hi} - i_{lo}$; if $i_{lo} = 1$ and $i_{hi} = n$, the number is approximately $\frac{10}{3}n^3$.

To form the orthogonal matrix Q this routine may be followed by a call to F08NFF (SORGHR/DORGHR):

```
CALL SORGHR (N, ILO, IHI, A, LDA, TAU, WORK, LWORK, INFO)
```

To apply Q to an m by n real matrix C this routine may be followed by a call to F08NGF (SORMHR/DORMHR). For example,

```
CALL SORMHR ('Left', 'No Transpose', M, N, ILO, IHI, A, LDA, TAU, C, LDC,
+          WORK, LWORK, INFO)
```

forms the matrix product QC .

The complex analogue of this routine is F08NSF (CGEHRD/ZGEHRD).

9 Example

To compute the upper Hessenberg form of the matrix A , where

$$A = \begin{pmatrix} 0.35 & 0.45 & -0.14 & -0.17 \\ 0.09 & 0.07 & -0.54 & 0.35 \\ -0.44 & -0.33 & -0.03 & 0.17 \\ 0.25 & -0.32 & -0.13 & 0.11 \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08NEF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5, NOUT=6)
INTEGER          NMAX, LDA, LWORK
PARAMETER       (NMAX=8, LDA=NMAX, LWORK=64*NMAX)
real           ZERO
PARAMETER       (ZERO=0.0e0)
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
*      .. Local Arrays ..
real           A(LDA, NMAX), TAU(NMAX-1), WORK(LWORK)
*      .. External Subroutines ..
EXTERNAL         sgehrd, X04CAF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08NEF Example Program Results'
Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
READ (NIN,*) ((A(I,J), J=1,N), I=1,N)
*
*      Reduce A to upper Hessenberg form
*
CALL sgehrd(N, 1, N, A, LDA, TAU, WORK, LWORK, INFO)
*
```

```

*      Set the elements below the first sub-diagonal to zero
*
      DO 40 I = 1, N - 2
        DO 20 J = I + 2, N
          A(J,I) = ZERO
20      CONTINUE
40      CONTINUE
*
*      Print upper Hessenberg form
*
      WRITE (NOUT,*)
      IFAIL = 0
*
      CALL X04CAF('General',' ',N,N,A,LDA,'Upper Hessenberg form',
+              IFAIL)
*
      END IF
      STOP
      END

```

9.2 Program Data

```

F08NEF Example Program Data
4                               :Value of N
0.35   0.45  -0.14  -0.17
0.09   0.07  -0.54   0.35
-0.44  -0.33  -0.03   0.17
0.25  -0.32  -0.13   0.11   :End of matrix A

```

9.3 Program Results

F08NEF Example Program Results

```

Upper Hessenberg form
      1      2      3      4
1  0.3500 -0.1160 -0.3886 -0.2942
2 -0.5140  0.1225  0.1004  0.1126
3  0.0000  0.6443 -0.1357 -0.0977
4  0.0000  0.0000  0.4262  0.1632

```
